* clean_enh_inactivos.do
* This file applies some labels to the ENH datasets and cleans some of the variables.
* Cleans only the "inactivos" files - Registro tipo 80
* Inputs: All files in C:/Users/JorgeEduardo/Dropbox/AA_Minimum_Wage/Data/Importable/ENH
* Outputs: All files in C:/Users/JorgeEduardo/Dropbox/AA_Minimum_Wage/Data/Clean/ENH

clear all
version 13.1

* Project info
project, doinfo   
local master "`r(pdir)'"
local doname "`r(dofile)'" 

* Directory
cd "../../Data/Importable/ENH"

* Project calls
project, original("../../../Code/Clean/labels_enh_inactivos.do")


* Registro tipo 80 - Inactivos

* Obtener una lista de variables comunes en todas las bases de datos
forvalues year=1996(1)2000 {
	forvalues trim=1(1)4 {
		use "`year'/TRIM `trim'/`year'_`trim'_inactivos.dta" , clear
		des, s
		* cap noi codebook salario_period
		unab vars: *
		if `year'==1996 & `trim'==1 local cv="`vars'"
		else local cv : list cv & vars
	}
}
di "`cv'" 

* tipo_registro id municipio segmento estrato sector seccion manzana vivienda hogar personas_hogar resultado_encuesta orden tiempo_inactivo ha_buscado cuando_busco area fex


quietly {
	forvalues year=1996(1)2000 {
		forvalues trim=1(1)4 {
			project, uses("`year'/TRIM `trim'/`year'_`trim'_inactivos.dta")
			use "`year'/TRIM `trim'/`year'_`trim'_inactivos.dta" , clear
			do "../../../Code/Clean/labels_enh_inactivos.do"
			drop if id==""
			drop if orden==.
			*** Base specific problems
			* Blanks in identifier in 1997
			if `year'==1997  replace id=subinstr(id," ","0",.)
			* Problema con un id en 1999 -4
			if `year'==1999 & `trim'==4 replace id="7300103630040010800201101" if id=="730010363004 010800201101"
			* Hay un hogar que tiene 43! personas en 1997-1. No sé si preparan la comida todos juntos. Mejor lo elimino
			drop if personas_hogar==43 
			
			* Work on common variables
			la val tipo_registro tipo
			**
			cap confirm numeric variable tiempo_trabajando
			if !_rc replace tiempo_trabajando =. if tiempo_trabajando==9
			**
			cap confirm variable actividad_antes
			if !_rc {
				cap confirm string variable actividad_antes
				if !_rc destring actividad_antes, replace
				gen cactividad_antes = recode(actividad_antes,13,29,39,42,50,63,72,83,96,99)
				replace cactividad_antes = . if actividad_antes==99 | actividad_antes==0
				la val cactividad_antes cactividad_empresa
			}
			**
			cap confirm variable segunda_actividad
			if !_rc la val segunda_actividad segunda_actividad
			**
			la val tiempo_inactivo tiempo_inactivo
			**
			cap confirm variable motivo_dejo_trabajo
			if !_rc {
				la val motivo_dejo_trabajo motivo_dejo_trabajo
				replace motivo_dejo_trabajo = . if motivo_dejo_trabajo==0
			}
			**
			la val ha_buscado sino
			**
			cap confirm string variable cuando_busco
			if !_rc destring cuando_busco, replace
			*Codificación de área está diferente dentro de los años, ¡Qué sufrimiento!
			replace area=76 if area==761
			replace area=11 if area==111
			if `year'==1996 | `year'==1997 & `trim'<=2 {
				gen area961_972=area
				merge n:1 area961_972 using "../../xwalks/area.dta", update replace keep(1 3 4 5)
				drop _merge nombre
			}			
			la val area area
			noi des, s
			noi codebook, detail problems all
			* noi codebook, problems all
			compress 
			save "../../Clean/ENH/`year'/TRIM `trim'/`year'_`trim'_inactivos.dta", replace
			project, creates("../../Clean/ENH/`year'/TRIM `trim'/`year'_`trim'_inactivos.dta")
		}
	}
}
  